
Abstract. The OTS/CafeOBJ method is an instance of the proof score
approach to systems analysis, which has been mainly devoted by re-
searchers in the OBJ community. We describe some tips on writing proof
scores in the OTS/CafeOBJ method and use a mutual exclusion proto-
col to exemplify the tips. We also argue soundness of proof scores in the
OTS/CafeOBJ method.

1 Introduction

The proof score approach to systems analysis has been mainly devoted by re-
searchers in the OBJ community [10, 8]. In the approach, an executable algebraic
specification language is used to specify systems and system properties, and a
processor of the language, which has a rewrite engine as one of its functionalities,
is used as a proof assistant to prove that systems satisfy system properties. Proof
plans called proof scores are written in the algebraic specification language to
conduct such proofs and the proof scores are executed by the language processor
by means of rewriting to check if the proofs are success.

Proof scores can be regarded as programs to prove that algebraic specifi-
cations satisfy system properties. While proof scores are being designed, con-
structed and debugged, we can understand algebraic specifications being ana-
lyzed more profoundly, which may even let us find flaws lurked in the specifica-
tions [15, 14]. Our thought on proof is similar to that of the designers of LP [11].
Proof scripts written in a tactic language provided by proof assistants such as
Coq [1] and Isabel/HOL [13] may be regarded as such programs, but it seems
that such proof assistants rather aim for mechanizing mathematics.

We have argued that the proof score approach to systems analysis is an
attractive approach to design verification in [6] thanks to (1) balanced human-
computer interaction and (2) flexible but clear structure of proof scores. The
former means that humans are able to focus on proof plans, while tedious and
detailed computations can be left to computers; humans do not necessarily have
to know what deductive rules or equations should be applied to goals to prove.
The latter means that lemmas do not need to be proved in advance and proof
scores can help humans comprehend the corresponding proofs; a proof that a



system satisfies a system property can be conducted even when all lemmas used
have not been proved, and assumptions used are explicitly and clearly written
in proof scores. To precisely assess the achievement of (1) and (2) in the proof
score approach and compare it with systems analysis with other existing proof
assistants, however, we need further studies.

The OTS/CafeOBJ method [17, 4, 7] is an instance of the proof score ap-
proach to systems analysis. In the OTS/CafeOBJ method, observational tran-
sition systems (OTSs) are used as models of systems and CafeOBJ [2], an ex-
ecutable algebraic specification language/system, is used; OTSs are transition
systems, which are straightforwardly written as algebraic specifications. An older
version of the OTS/CafeOBJ method is described in [17, 4], and the latest version
is described in [7]. We have conducted case studies, among which are [15, 18, 19,
16, 20], to demonstrate the usefulness of the OTS/CafeOBJ method. In this pa-
per, we describe some tips on writing proof scores in the OTS/CafeOBJ method.
A mutual exclusion protocol called Tlock using atomicInc, which atomically in-
crements the number stored in a variable and returns the old number, is used
as an example. We also argue soundness of proof scores in the OTS/CafeOBJ
method.

The rest of the paper is organized as follows. Section 2 describes the
OTS/CafeOBJ method. Section 3 describes tips on writing proof scores in the
OTS/CafeOBJ method. Section 4 informally argue soundness of proof scores in
the OTS/CafeOBJ method. Section 5 concludes the paper.

2 The OTS/CafeOBJ Method

In the OTS/CafeOBJ method, systems are analyzed as follows.

1. Model a system as an OTS S.
2. Write S in CafeOBJ as an algebraic specification. The specification consists

of sorts (or types), operators on the sorts, and equations that define (proper-
ties of) the operators. The specification can be executed by using equations
as left-to-right rewrite rules by CafeOBJ.

3. Write system properties in CafeOBJ. Let P be the set of such system prop-
erties and let P ′ be the empty set. .

4. If P is empty, the analysis has been successfully finished, which means that
S satisfies all the properties in P ′. Otherwise, extract a property p from P
and go next. .

5. Write a proof score in CafeOBJ to prove that S satisfies p. The proof may
need other system properties as lemmas. Write such system properties in
CafeOBJ and put them that are not in P ′ into P if any. .

6. Execute (or play) the proof score with CafeOBJ. If all the results are as
expected, then the proof is discharged. Put p into P ′ and go to 4. If all the
results are not as expected, rewrite the proof score and repeat 6. .

Tasks 5 and 6 may be interactively conducted together. A counterexample may
be found in tasks 5 and 6.



In this section, we mention CafeOBJ, describe the definitions of basic con-
cepts on OTSs, write on how to write OTSs in CafeOBJ and how to write proof
scores that OTSs satisfy invariant properties in CafeOBJ.

2.1 CafeOBJ

CafeOBJ [2] is an algebraic specification language/system mainly based on order-
sorted algebras and hidden algebras [9, 3]. Abstract data types are specified in
terms of order-sorted algebras, and abstract machines are specified in terms of
hidden algebras. Algebraic specifications of abstract machines are called behav-
ioral specifications. There are two kinds of sorts in CafeOBJ: visible sorts and
hidden sorts. A visible sort denotes an abstract data type, while a hidden sort
denotes the state space of an abstract machine. There are three kinds of opera-
tors (or operations) with respect to (wrt) hidden sorts: hidden constants, action
operators and observation operators. Hidden constants denote initial states of ab-
stract machines, action operators denote state transitions of abstract machines,
and observation operators let us know the situation where abstract machines are
located. Both an action operator and an observation operator take a state of an
abstract machine and zero or more data. The action operator returns the suc-
cessor state of the state wrt the state transition denoted by the action operator
plus the data. The observation operator returns a value that characterizes the
situation where the abstract machine is located.

Basic units of CafeOBJ specifications are modules. CafeOBJ provides built-in
modules. One of the most important built-in modules is BOOL in which proposi-
tional logic is specified. BOOL is automatically imported by almost every module
unless otherwise stated. In BOOL and its parent modules, declared are the visible
sort Bool, the constants true and false of Bool, and operators denoting some
basic logical connectives. Among the operators are not_, _and_, _or_, _xor_,
_implies_ and _iff_ denoting negation (¬), conjunction (∧), disjunction (∨),
exclusive disjunction (xor), implication (⇒) and logical equivalence (⇔), re-
spectively. The operator if_then_else_fi corresponding to the if construct in
programming languages is also declared. CafeOBJ uses the Hsiang term rewrit-
ing system (TRS) [12] as the decision procedure for propositional logic, which is
implemented in BOOL. CafeOBJ reduces any term denoting a proposition that is
always true (false) to true (false). More generally, a term denoting a proposi-
tion reduces to an exclusively disjunctive normal form of the proposition.

2.2 Observational Transition Systems (OTSs)

We suppose that there exists a universal state space denoted Υ and that each
data type used in OTSs is provided. The data types include Bool for truth values.
A data type is denoted D∗.

Definition 1 (OTSs). An OTS S is 〈O, I, T 〉 such that

– O : A finite set of observers. Each observer ox1:Do1,...,xm:Dom
: Υ → Do

is an indexed function that has m indexes x1, . . . , xm whose types are



Do1, . . . , Dom. The equivalence relation (υ1 =S υ2) between two states
υ1, υ2 ∈ Υ is defined as ∀ox1,...,xm : O. (ox1,...,xm(υ1) = ox1,...,xm(υ2)), where
∀ox1,...,xm : O is the abbreviation of ∀ox1,...,xm : O.∀x1 : Do1 . . .∀xm : Dom.

– I : The set of initial states such that I ⊆ Υ .
– T : A finite set of transitions. Each transition ty1:Dt1,...,yn:Dtn : Υ → Υ is an

indexed function that has n indexes y1, . . . , yn whose types are Dt1, . . . , Dtn

provided that ty1,...,yn
(υ1) =S ty1,...,yn

(υ2) for each [υ] ∈ Υ/=S , each υ1, υ2 ∈
[υ] and each yk : Dtk for k = 1, . . . , n. ty1,...,yn

(υ) is called the successor state
of υ wrt S. Each transition ty1,...,yn

has the condition c-ty1:Dt1,...,yn:Dtn
: Υ →

Bool, which is called the effective condition of the transition. If c-ty1,...,yn(υ)
does not hold, then ty1,...,yn(υ) =S υ. �

We note the following two points on transitions, which have something to
do with writing proof scores. (1) Although transitions are defined as relations
among states in some other existing transition systems, transitions are func-
tions on states in OTSs. This is because transitions are represented by (action)
operators in behavioral specifications and operators are functions in CafeOBJ.
However, multiple transitions that are functions on states can be substituted for
one transition that is a relation among states. (2) Basically there is no restric-
tion on the form of effective conditions. But, effective conditions should be in
the form c1-ty1,...,yn(υ) ∧ . . . ∧ cM -ty1,...,yn(υ), where each ck-ty1,...,yn(υ) has no
logical connectives or has one negation at head, so that proof scores can have
clear structure. When an effective condition is not in this form, it is converted to
a disjunctive normal form. If the disjunctive normal form has more than one dis-
junct, multiple transitions each of which has one of the disjuncts as its effective
condition can be substituted for the corresponding transition.

Definition 2 (Reachable states). Given an OTS S, reachable states wrt S
are inductively defined:

– Each υinit ∈ I is reachable wrt S.
– For each ty1,...,yn ∈ T and each yk : Dtk for k = 1. . . . , n, tx1,...,xn(υ) is

reachable wrt S if υ ∈ Υ is reachable wrt S.

Let RS be the set of all reachable states wrt S . �

Predicates whose types are Υ → Bool are called state predicates. All proper-
ties considered in this paper are invariants.

Definition 3 (Invariants). Any state predicate p : Υ → Bool is called invari-
ant wrt S if p holds in all reachable states wrt S, i.e. ∀υ : RS . p(υ). �

We suppose that each state predicate p considered in this paper has the form
∀z1 : Dp1 . . .∀za : Dpa. P (υ, z1, . . . , za), where υ, z1, . . . , za are all variables in p
and P (υ, z1, . . . , za) does not contain any quantifiers.

A concrete example of how to model a system as an OTS is given.

Example 1 (Tlock). The pseudo-code executed by each process i can be written
as follows:



Loop
l1: ticket[i] := atomicInc(tvm);
l2: repeat until ticket[i] = turn;

Critical section;
cs: turn := turn + 1;

tvm and turn are non-negative integer variables shared by all processes and
ticket[i] is a non-negative integer variable that is local to process i. Initially,
each process i is at label l1, tvm and turn are 0, and ticket[i] for each i is
unspecified. The value of tvm (which stands for a ticket vending machine) is the
next available ticket. Each process i obtains a ticket, which is stored in ticket[i],
at label l1. A process is allowed to enter the critical section if its ticket equals the
value of turn at label l2. turn is incremented when a process leaves the critical
section at label cs.

Let Label, Pid and Nat be the types of labels (l1, l2 and cs), process IDs
and non-negative integers (natural numbers). Tlock can be modeled as the OTS
STlock such that

– OTlock , {tvm : Υ → Nat, turn : Υ → Nat, ticketi:Pid : Υ → Nat, pci:Pid : Υ →
Label}

– ITlock , {υinit ∈ Υ | tvm(υinit) = 0 ∧ turn(υinit) = 0 ∧ ∀i : Pid. (pci(υinit) = l1)}
– TTlock , {geti:Pid : Υ → Υ, checki:Pid : Υ → Υ, exiti:Pid : Υ → Υ}

The three transitions are defined as follows:

– geti : c-geti(υ) , pci(υ) = l1. If c-wanti(υ), then

tvm(geti(υ)) , tvm(υ) + 1, turn(geti(υ)) , turn(υ),

ticketj(geti(υ)) , if i = j tvm(υ) elseticketj(υ), and

pcj(geti(υ)) , if i = j then l2 else pcj(υ).

– checki : c-checki(υ) , pci(υ) = l2 ∧ ticketi(υ) = turn(υ). If c-wanti(υ), then

tvm(checki(υ)) , tvm(υ), turn(checki(υ)) , turn(υ),

ticketj(geti(υ)) , ticketj(υ), and pcj(checki(υ)) , if i = j then cs else pcj(υ).

– exiti : c-exiti(υ) , pci(υ) = cs. If c-wanti(υ), then

tvm(exiti(υ)) , tvm(υ) + 1, turn(exiti(υ)) , turn(υ),

ticketj(geti(υ)) , ticketj(υ), and pcj(exiti(υ)) , if i = j then l1 else pcj(υ).

Let MX(υ) be ∀i, j : Pid. [(pci(υ) = cs ∧ pcj(υ) = cs) ⇒ i = j]. MX(υ)
is invariant wrt STlock, i.e. ∀υ : RSTlock .MX(υ), although it may need to be
verified. �

2.3 Specifying OTSs in CafeOBJ

We suppose that a visible sort V∗ corresponding to each data type D∗ used in
OTSs and the related operators are provided. Xk and Yk are CafeOBJ variables
corresponding to indexes xk and yk of observers and transitions, respectively.



The universal state space Υ is represented by a hidden sort, say H declared
as *[ H ]* by enclosing it with *[ and ]*. Given an OTS S, an arbitrary initial
state is represented by a hidden constant, say init, each observer ox1,...,xm

is represented by an observation operator, say o, and each transition ty1,...,yn

is represented by an action operator, say t. The hidden constant init, the
observation operator o and the action operator t are declared as follows:

op init : -> H
bop o : H Vo1 ...Vom -> Vo

bop t : H Vt1 ...Vtn -> H

The keyword bop or bops is used to declare observation and action operators.
We suppose that the value returned by ox1,...,xm

in an arbitrary initial state
can be expressed as f(x1, . . . , xm). This is expressed by the following equation:

eq o(init,X1,...,Xm) = f(X1,...,Xm) .

f(X1,...,Xm) is the CafeOBJ term corresponding to f(x1, . . . , xm).
Each transition ty1,...,yn

is defined by describing what the value returned by
each observer ox1,...,xm

in the successor state becomes when ty1,...,yn
is applied in

a state υ. When c-ty1,...,yn(υ) holds, this is expressed generally by a conditional
equation that has the form

ceq o(t(S,Y1,...,Yn),X1,...,Xm) = e-t(S,Y1,...,Yn,X1,...,Xm)
if c-t(S,Y1,...,Yn) .

S is a CafeOBJ variable of H, corresponding to υ. e-t(S,Y1,...,Yn,X1,...,Xm)
is the CafeOBJ term corresponding to the value returned by ox1,...,xm in the
successor state denoted by t(S,Y1,...,Yn). c-t(S,Y1,...,Yn) is the CafeOBJ
term corresponding to c-ty1,...,yn

(υ).
If c-ty1,...,yn

(υ) always holds in any state υ or the value returned by ox1,...,xm

is not affected by applying ty1,...,yn
in any state υ (i.e. regardless of the truth

value of c-ty1,...,yn(υ)), then a usual equation is used instead of a conditional
equation. The usual equation has the form

eq o(t(S,Y1,...,Yn),X1,...,Xm) = e-t(S,Y1,...,Yn,X1,...,Xm) .

e-t(S,Y1,...,Yn,X1,...,Xm) is S if the value returned by ox1,...,xm
is not

affected by applying ty1,...,yn
in any state.

When c-ty1,...,yn(υ) does not hold, ty1,...,yn changes nothing, which is ex-
pressed by a conditional equation that has the form

ceq t(S,Y1,...,Yn) = S if not c-t(S,Y1,...,Yn) .

We give the CafeOBJ specification of STlock.

Example 2 (CafeOBJ specification of STlock). SQlock is specified in CafeOBJ as
the module TLOCK:



mod* TLOCK { pr(PNAT) pr(LABEL) pr(PID)

*[Sys]*

-- an arbitrary initial state

op init : -> Sys

-- observation operators

bops tvm turn : Sys -> Nat bop ticket : Sys Pid -> Nat

bop pc : Sys Pid -> Label

-- action operators

bops get check exit : Sys Pid -> Sys

-- CafeOBJ variables

var S : Sys vars I J : Pid

-- init

eq tvm(init) = 0 . eq turn(init) = 0 . eq pc(init,I) = l1 .

-- get

op c-get : Sys Pid -> Bool {strat: (0 1 2)}

eq c-get(S,I) = (pc(S,I) = l1) .

--

ceq tvm(get(S,I)) = s(tvm(S)) if c-get(S,I) .

eq turn(get(S,I)) = turn(S) .

ceq ticket(get(S,I),J)

= (if I = J then tvm(S) else ticket(S,J) fi) if c-get(S,I) .

ceq pc(get(S,I),J) = (if I = J then l2 else pc(S,J) fi) if c-get(S,I) .

ceq get(S,I) = S if not c-get(S,I) .

-- check

op c-check : Sys Pid -> Bool {strat: (0 1 2)}

eq c-check(S,I) = (pc(S,I) = l2 and ticket(S,I) = turn(S)) .

--

eq tvm(check(S,I)) = tvm(S) . eq turn(check(S,I)) = turn(S) .

eq ticket(check(S,I),J) = ticket(S,J) .

ceq pc(check(S,I),J)

= (if I = J then cs else pc(S,J) fi) if c-check(S,I) .

ceq check(S,I) = S if not c-check(S,I) .

-- exit

op c-exit : Sys Pid -> Bool {strat: (0 1 2)}

eq c-exit(S,I) = (pc(S,I) = cs) .

--

eq tvm(exit(S,I)) = tvm(S) .

ceq turn(exit(S,I)) = s(turn(S)) if c-exit(S,I) .

eq ticket(exit(S,I),J) = ticket(S,J) .

ceq pc(exit(S,I),J)

= (if I = J then l1 else pc(S,J) fi) if c-exit(S,I) .

ceq exit(S,I) = S if not c-exit(S,I) .

}

A comment starts with -- and terminates at the end of the line. PNAT, LABEL
and PID are the modules in which natural numbers, labels and process IDs
are specified. The keyword pr is used to imports modules. The operator s of
s(tvm(S)) and s(turn(S)) is the successor function of natural numbers. The
keyword start: is used to specify local strategies to operators [5]. The local
strategy (0 1 2) given to c-get indicates that when CafeOBJ meets a term



whose top is c-get such as c-get(s,i), CafeOBJ should try to rewrite the
whole term such as c-get(s,i). If CafeOBJ does not find any rules with which
the term is rewritten, it evaluates the first and second arguments such as s and
i in that order, and tries to rewrite the whole term such as c-get(s′,i′) again,
where s′ and i′ are the results obtained by evaluating s and i. �

2.4 Proof Scores of Invariants

Although some invariants may be proved by rewriting and/or case splitting only,
we often need to use induction, especially simultaneous induction [7]. We then
describe how to verify ∀υ : RS . p(υ) by simultaneous induction by writing proof
scores in CafeOBJ based on the CafeOBJ specification of S.

It is often impossible to prove ∀υ : RS . p(υ) alone. We then suppose that it is
possible to prove ∀υ : RS . p(υ) together with N −1 other state predicates3, that
is, we prove ∀υ : RS . (p1(υ) ∧ . . . ∧ pN (υ)), where p1 is p. We suppose that each
pk has the form ∀zk : Dpk. Pk(υ, zk) for k = 1, . . . , N . Note that the method
described here can be used when pk has more than one universally quantified
variable. Let υc

init be an arbitrary initial state of S, and then for the base case,
all we have to do is to prove

∀z1 : Dp1. P1(υc
init, z1) ∧ . . . ∧ ∀zN : DpN . PN (υc

init, zN ) (1)

For each induction case (i.e. each ty1,...,yn ∈ T ), all we have to do is to prove

∀z1 : Dp1. P1(υc, z1) ∧ . . . ∧ ∀zN : DpN . PN (υc, zN )
⇒ ∀z1 : Dp1. P1(tyc

1,...,yc
n
(υc), z1) ∧ . . . ∧ ∀zN : DpN . PN (tyc

1,...,yc
n
(υc), zN ) (2)

for an arbitrary state υc and an arbitrary value yc
k for k = 1, . . . , n.

To prove (1), we can separately prove each conjunct

Pi(υc
init, z

c
k) (3)

where zc
k is an arbitrary value of Dpk for k = 1, . . . , N . To prove (2), assuming

∀z1 : Dp1. P1(υc, z1), . . . , ∀zN : DpN . PN (υc, zN ), we can separately prove each
Pk(tyc

1,...,yc
n
(υc), zc

k), where zc
k is an arbitrary value of Dpk, for k = 1, . . . , N .

Pk(υc, zc
k) is often used as an assumption to prove Pk(tyc

1,...,yc
n
(υc), zc

k). Therefore,
the formula to prove has the form

(Pα(υc, dα) ∧ Pβ(υc, dβ) ∧ . . .) ⇒ [Pk(υc, zc
k) ⇒ Pk(tyc

1,...,yc
n
(υc), zc

k)] (4)

where α, β, . . . ∈ {1, . . . , N} and dα, dβ , . . . are some values of Dpα, Dpβ , . . . for
i = 1, . . . , N .

We next describe how to write proof plans of (3) and (4) in CafeOBJ. We
first declare the operators denoting P1, . . . , PN and the equations defining the
operators. The operators and equations are declared in a module, say INV (which
imports the module where S is written), as follows:
3 Generally, such N − 1 state predicates should be found while ∀υ : RS . p(υ) is being

proved.



op invk : H Vpk -> Bool
eq invk(S,Zk) = Pi(S,Zk) .

for k = 1, . . . , N . Zk is a CafeOBJ variable of Vpk and Pi(S,Zk) is a CafeOBJ term
denoting Pk(υ, zk). In INV, we also declare a constant zc

k denoting an arbitrary
value of Vpk for i = 1, . . . , N . We then declare the operators denoting basic
formulas to prove in the induction cases and the equations defining the operators.
The operators and equations are declared in a module, say ISTEP (which imports
INV), as follows:

op istepk : Vpk -> Bool
eq istepk(Zk) = invk(s,Zk) implies invk(s’,Zk) .

for i = 1, . . . , N . s and s’, which are declared in ISTEP, are constants of H. s
denotes an arbitrary state and s’ denotes a successor state of the state.

The proof plan of (3), written in CafeOBJ, has the form

open INV
red invk(init,zc

k) .
close

for i = 1, . . . , N . The command open makes a temporary module that imports
a given module and the command close destroys it. The command red reduces
a given term. CafeOBJ scripts like this constitute proof scores. Such fragments
of proof scores are called proof passages. Feeding such a proof passage into the
CafeOBJ system, if the CafeOBJ system returns true, the corresponding proof
is successfully done.

The proof of (4) often needs case splitting. We suppose that the state space
is split into Lk sub-spaces4 in order to prove (4) and that each sub-space is
characterized by a proposition casekl for l = 1, . . . , Lk provided that casek1 ∨
. . . ∨ casekLk

. The proof of (4) can be then replaced with

casekl ⇒
[(Pα(υc, dα) ∧ Pβ(υc, dβ) ∧ . . .) ⇒ [Pk(υc, zc

k) ⇒ Pk(tyc
1,...,yc

n
(υc), zc

k)]] (5)

for l = 1, . . . , Lk and k = 1, . . . , N .
We suppose that dα, dβ , . . . are CafeOBJ terms denoting dα, dβ , . . . Then the

proof passage of (5) has the form

open ISTEP
-- arbitrary objects
op yc

1 : -> V1 . · · · op yc
N : -> VN .

-- assumptions
Declaration of equations denoting casekl.
-- successor state
eq s’ = t(s,yc

1,...,y
c
N) .

-- check
red (invα(s,dα) and invβ(s,dβ) and ...) implies istepk(zc

k) .
close

4 Generally, such case splitting should be done while ∀υ : RS . p(υ) is being proved.



for l = 1, . . . , Lk and k = 1, . . . , N .
Equations available in a proof passage “open M · · · close” are those declared

in the module M and the modules imported by M plus those declared in the
proof passage. We say that the lefthand side of an equation l = r (a term t) is
(ir)reducible in a proof passage if l (t) is (ir)reducible wrt E \{l = r} (E), where
E is the set of all equations available in the proof passage.

We briefly describe the proof scores of ∀υ : RSTlock .MX(υ).

Example 3 (Proof socres of ∀υ : RSTlock .MX(υ)). We need four more state pred-
icates to prove ∀υ : RSTlock .MX(υ), which are found while proving it. The four
state predicates are as follows: p2(υ) , ∀i, j : Pid. [(pci(υ) = cs ∧ pcj(υ) =
l2 ∧ ticketj(υ) = turn(υ)) ⇒ i = j], p3(υ) , ∀i : Pid. (pci(υ) = cs ⇒
turn(υ) < tvm(υ)), p4(υ) , ∀i, j : Pid. [(pci(υ) = l2 ∧ pcj(υ) = l2 ∧ ticketi(υ) =
ticketj(υ) ⇒ i = j], and p5(υ) , ∀i : Pid. (pci(υ) = l2 ⇒ ticketi(υ) < tvm(υ)).
The proof of ∀υ : RSTlock .MX(υ) needs p2, that of ∀υ : RSTlock . p2(υ) needs MX,
p3 and p4, that of ∀υ : RSTlock . p3(υ) needs MX and p5, that of ∀υ : RSTlock . p4(υ)
needs p5, and that of ∀υ : RSTlock . p5(υ) needs no other state predicates.

The module INV is declared as follows:

mod INV { pr(TLOCK)

ops i j : -> Pid

op inv1 : Sys Pid Pid -> Bool op inv2 : Sys Pid Pid -> Bool

op inv3 : Sys Pid -> Bool op inv4 : Sys Pid Pid -> Bool

op inv5 : Sys Pid -> Bool

var S : Sys vars I J : Pid

eq inv1(S,I,J) = ((pc(S,I) = cs and pc(S,J) = cs) implies I = J) .

eq inv2(S,I,J) = ((pc(S,I) = cs and pc(S,J) = l2

and ticket(S,J) = turn(S)) implies I = J) .

eq inv3(S,I) = (pc(S,I) = cs implies turn(S) < tvm(S)) .

eq inv4(S,I,J) = ((pc(S,I) = l2 and pc(S,J) = l2

and ticket(S,I) = ticket(S,J)) implies I = J) .

eq inv5(S,I) = (pc(S,I) = l2 implies ticket(S,I) < tvm(S)) .

}

The module ISTEP is declared as follows:

mod ISTEP { pr(INV)

ops s s’ : -> Sys

op istep1 : Pid Pid -> Bool op istep2 : Pid Pid -> Bool

op istep3 : Pid -> Bool op istep4 : Pid Pid -> Bool

op istep5 : Pid -> Bool

vars I J : Pid

eq istep1(I,J) = inv1(s,I,J) implies inv1(s’,I,J) .

eq istep2(I,J) = inv2(s,I,J) implies inv2(s’,I,J) .

eq istep3(I) = inv3(s,I) implies inv3(s’,I) .

eq istep4(I,J) = inv4(s,I,J) implies inv4(s’,I,J) .

eq istep5(I) = inv5(s,I) implies inv5(s’,I) .

}



Let us consider the following proof passage of ∀υ : RSTlock .MX(υ):

open ISTEP

-- arbitrary values

op k : -> Pid .

-- assumptions

-- eq c-check(s,k) = true .

eq pc(s,k) = l2 . eq ticket(s,k) = turn(s) .

qq i = k . eq (j = k) = false . eq pc(s,j) = cs .

-- successor state

eq s’ = check(s,k) .

-- check

red istep1(i,j) .

close

The proof passage corresponds to a (sub-)case obtained by splitting the induc-
tion case for checkk. The (sub-)case is referred as case 1.check.1.1.0.1. CafeOBJ
returns false for the proof passage. From the five equations that character-
ize the (sub-)case, however, we can conjecture p2. When inv2(s,j,i) implies
istep1(i,j) is used instead of istep1(i,j), CafeOBJ returns true for the
proof passage.

Let us consider the following proof passage of ∀υ : RSTlock . p2(υ):

open ISTEP

-- arbitrary values

op k : -> Pid .

-- assumptions

-- eq c-exit(s,k) = true .

eq pc(s,k) = cs .

eq (i = k) = false . eq (j = k) = false . eq pc(s,i) = cs .

-- successor state

eq s’ = exit(s,k) .

-- check

red istep2(i,j) .

close

The proof passage corresponds to a (sub-)case obtained by splitting the in-
duction case for exitk. The (sub-)case is referred as case 2.exit.1.0.0.1. Al-
though CafeOBJ returns neither true nor false for the proof passage, we
notice that inv1(s,i,k) reduces to false in the proof passage. Therefore,
we use inv1(s,i,k) implies istep2(i,j) instead of istep2(i,j) and then
CafeOBJ returns true for the proof passage. �

3 Tips

What we should do to prove a state predicate invariant wrt an OTS is three
tasks: (1) use of simultaneous induction, (2) case splitting and (3) predicate
(lemma) discovery/use. We use the proof of ∀υ : RSTlock .MX(υ) to describe the
three tasks.



3.1 Simultaneous Induction

The first thing to do is to use simultaneous induction to break the proof into
the four (sub-)goals (one is the base case and the others are the three induction
cases) and the four proof passages are written. The proof passage of the base
case is as follows:

open INV

red inv1(init,i,j) .

close

The proof passage of the induction case for checkk is as follows:

open ISTEP

op k : -> Pid .

eq s’ = check(s,k) .

red istep1(i,j) .

close

The case is referred as case 1.check. The proof passages of the remaining two
induction cases are written likewise.

CafeOBJ returns true for the base case but neither true nor false for each
of the three induction cases. What to do for the three induction cases are case
splitting and/or predicate discovery/use.

3.2 First Thing to Do for Each Induction Case

Each induction case for ty1,...,yn is split into two (sub-)cases: (1) c-ty1,...,yn and
(2) ¬c-ty1,...,yn unless c-ty1,...,yn holds in every case. Case 1.check is split into the
two (sub-)cases whose corresponding proof passages are as follows:

open ISTEP

op k : -> Pid .

eq c-check(s,k) = true .

eq s’ = check(s,k) .

red istep1(i,j) .

close

open ISTEP

op k : -> Pid .

eq c-check(s,k) = false .

eq s’ = check(s,k) .

red istep1(i,j) .

close

The two (sub-)cases are referred as case 1.check.1 and 1.check.0. CafeOBJ re-
turns true for case 1.check.0 but neither true nor false for case 1.check.1.
CafeOBJ always returns true for the (sub-)case where ¬c-ty1,...,yn

due to Defi-
nition 1 if the OTS concerned is correctly written in CafeOBJ.

3.3 Appropriate Equations Declared in Proof Passages

As shown, each (sub-)case is characterized by equations. Equational reasoning by
rewriting is used to check if a proposition holds in each case, but full equational
reasoning power is not used because CafeOBJ does not employ any completion
facilities. Therefore, equations that characterize a case heavily affects the success



in proving that a proposition holds in the case. We describe appropriate equa-
tions, which characterize a case, declared in a proof passage. If CafeOBJ returns
true for a proof passage, nothing should be done. Otherwise, the equations in
the proof passage should be appropriate as described from now.

– The lefthand side of each equation should be irreducible in a proof passage so
that the equation can be used effectively as a rewrite rule. This is because the
rewriting strategy adopted by CafeOBJ is basically an innermost strategy.

– Let PP(E), where E is a set of equations, be a proof passage in which
the equations in E are declared, and E1 and E2 be sets of equations. We
suppose that

∧
e1∈E1

e1 is equivalent to
∧

e2∈E2
e2. If every equation in E1

can be proved by rewriting from PP(E2) but every equation in E2 cannot
be proved by rewriting from PP(E1), then E2 should be used instead of E1.
Some examples are given.
1. Let E1 be {ρ1 ∧ ρ2 = true} and E2 be {ρ1 = true, ρ2 = true}. We

suppose that ρ1 ∧ ρ2, ρ1 and ρ2 are irreducible in PP(∅). Then, ρ1 ∧ ρ2

reduces to true in PP(E2) but l1 (l2) does not necessarily reduce to
true in PP(E1). Therefore, E2 should be used instead of E1.

2. Let c be a binary data constructor. We suppose that c(a1, b1) equals
c(a2, b2) if and only if a1 equals a2 and b1 equals b2. Let E1 be {c(a1, b1) =
c(a2, b2)} and E2 be {a1 = a2, b1 = b2}. We suppose that c(a1, b1), a1

and b1 are irreducible in PP(∅). Then, both c(a1, b1) and c(a2, b2) reduce
to a same term in PP(E2) but a1 and a2 (b1 and b2) do not necessarily
reduce to a same term in PP(E1). Therefore, E2 should be used instead
of E1.

3. Let n be a natural number, N be a constant denoting an arbitrary
multiset of natural numbers, the juxtaposition operator be a data
constructor of multisets. The juxtaposition operator is declared as
op __ : Bag Bag -> Bag {assoc comm id: empty}, where Bag is the
visible sort for multisets of natural numbers and is a supersort of Nat,
assoc and comm specify that the operator is associative and commuta-
tive, and id: empty specifies that empty, which is the constant denoting
the empty multiset, is an identity of the operator. We suppose that we
want to specify that N includes n. One way is to use n ∈ N = true, and
the other way is to use N = n N ′, where N ′ is another constant denoting
an arbitrary multiset of natural numbers5. Let E1 be {n ∈ N = true}
and E2 be {N = n N ′}. We suppose that n ∈ N and N are irreducible
in PP(∅). Then, n ∈ N reduces to true in PP(E2) if ∈ is defined ap-
propriately in equation, but N and n N ′ do not necessarily reduce to a
same term in PP(E1). Therefore, E2 should be used instead of E1.

4. ¬ρ is reducible in any proof passage because of the Hsiang TRS. If ρ
is irreducible in a proof passage, ¬ρ reduces to ρ xor true in the proof

5 Since N is an arbitrary multiset and includes n, N must be n′ N ′, where (1) n′

equals n or (2) n ∈ N ′. We can select (1) because the juxtaposition operator is
associative and commutative.



passage. Therefore, one way of making the equation (¬ρ) = true effective
is to use (ρ xor true) = true. But, ρ = false is more appropriate.

5. This example is a variant. Let E1 be {(l = r) = true} and E2 be {l = r}.
We suppose that l = r and l are irreducible in PP(∅). l = r reduces to
true in both PP(E1) and PP(E2). It is often the case, however, that E2

is more appropriate than E1 because l reduces r in PP(E2) but l does
not in PP(E1).

According to what has been described in this subsection, the proof passage
of case 1.check.1 should be rewritten as follows:

open ISTEP

op k : -> Pid .

-- eq c-check(s,k) = true .

eq pc(s,k) = l2 . eq ticket(s,k) = turn(s) .

eq s’ = check(s,k) .

red istep1(i,j) .

close

CafeOBJ still returns neither true nor false for this proof passage. Then, what
we should do is further case splitting.

3.4 Further Case Splitting

For a proof passage for which CafeOBJ returns neither true nor false, the case
corresponding to the proof passage is split into multiple (sub-)cases in each of
which CafeOBJ returns either true or false. When CafeOBJ returns true in a
(sub-)case, nothing should be done for the case. When CafeOBJ returns false
in a (sub-)case, it is necessary to find a state predicate that does not hold in the
case and is likely invariant wrt an OTS concerned.

There are some ways of splitting a case into multiple (sub-)cases.

– Based on a proposition ρ : A case is split into two (sub-)cases where (1) ρ
holds and (2) ρ does not, respectively. As shown in Subsect. 3.2, case 1.check
is split into the two (sub-)cases based on the proposition c-check(s,k).

– Based on data constructors : We suppose that a data type has M data con-
structors. Then, a case is split into M (sub-)cases. Some examples are given.
1. Nat has the two data constructors 0 and s. Let x be a constant denoting

an arbitrary natural number in a proof passage. The case corresponding
to the proof passage is split into the two (sub-)cases where (1) x = 0 and
(2) x = s(y), where y is another constant denoting an arbitrary natural
number. Case (1) means that x is zero and case (2) means that x is not
zero.

2. Bag has the two data constructors empty and __. Let N be a constant
denoting an arbitrary multiset in a proof passage. The case corresponding
to the proof passage is split into the two (sub-)cases where (1) N = empty
and (2) N = n′ N ′, where n′ is a constant denoting an arbitrary natural
number and N ′ is a constant denoting an arbitrary multiset. Case (1)
means that N is empty and case (2) means that N is not empty.



– Based on a tautology whose form is ρ1∨. . .∨ρM : A case is split into M (sub-
)cases where (1) ρ1 holds, . . . , (M) ρM holds. This case splitting generalizes
the case splitting based on a proposition because ρ ∨ ¬ρ is a tautology.

In order to apply one of the three ways of splitting a case, we need to find a
proposition, a constant denoting an arbitrary value of a data type, or a tautology
whose form is ρ1∨. . .∨ρM . There are usually multiple candidates based on which
a case is split. A selection from such candidates affects how well a proof concerned
is conducted. It is necessary to understand an OTS concerned and experience
writing proof scores so as to select a better one among such candidates. There
are some heuristic rules, however, to select one among such candidates.

– Select a proposition that directly affects the truth value of a proposition to
prove such as istep(i,j). If i equals j, istep(i,j) reduces to true in case
1.check.1, the proposition i = j may be a good candidate.

– Select a proposition ρ if ρ appears in a result obtained by reducing a propo-
sition to prove. If ρ appears at the conditional position of if_then_else_fi
such as if ρ then a else b fi, ρ may be a good candidate.

We describe how to split case 1.check.1. CafeOBJ returns ((if (k = i)
then cs else pc(s,i) fi) = cs) and ... for the corresponding proof pas-
sage. Then, we select the proposition k = i to split the case. The equation i =
k is declared6 in one proof passage whose corresponding case is referred as case
1.check.1.1, and the equation (i = k) = false is declared in the other proof
passage whose corresponding case is referred as case 1.check.1.0.

Since CafeOBJ returns if (k = j) then cs else pc(s,j) fi = cs and
... for the proof passage corresponding case 1.check.1.1, we select the propo-
sition k = j to split the case. The equation j = k is declared in one proof
passage whose corresponding case is referred as 1.check.1.1.1, and the equation
(j = k) = false is declared in the other proof passage whose corresponding
case is referred as case 1.check.1.1.0. CafeOBJ returns true for the former proof
passage, but pc(s,j) = cs xor true for the latter proof passage. Then, case
1.check.1.1.0 is also split based on pc(s,j) = cs. The equation pc(s,j) = cs
is declared in one proof passage whose corresponding case is referred as case
1.check.1.1.0.1, and the equation (pc(s,j) = cs) = false is declared in the
other proof passage whose corresponding case is referred as case 1.check.1.1.0.0.
CafeOBJ returns false for the former proof passage and true for the latter
proof passage. Case 1.check.1.0 can be split into four (sub-)cases in the same
was as case 1.check1.1.

3.5 Predicate (Lemma) Discovery/Use

When CafeOBJ returns false for a proof passage, there are two possibilities:
(1) if an an arbitrary state characterized by the case corresponding to the proof
passage is not reachable wrt an OTS S concerned, the case can be discharged,
6 Note that i = k is declared instead for k = i.



and (2) otherwise, a state predicate concerned is not invariant wrt S. If a state
predicate is invariant wrt S and does not hold in the case, then an arbitrary
state characterized by the case is not reachable wrt S. That is why we find a
state predicate that does not hold in the case and is likely invariant wrt S.

Let E is a set of equations that characterize a case such that CafeOBJ re-
turns false for a proof passage corresponding to the case. We suppose that∧

e∈E e is equivalent to a proposition whose form is Q(υc, zc
α). Let q(υ) be

∀zα : Dqα.¬Q(υ, zα). Since q surely does not hold in the case characterized by
E, q is one possible candidate. Generally, q′ such that q′ ⇒ q can be a candidate
because q′ does not hold in the case characterized by E,

Let us consider the proof passage corresponding to case 1.check.1.1.0.1
shown in Example 3. From the five equations that characterize the case, we
obtain the proposition pc(s,i) = l2 and pc(s,j) = cs and ticket(s,i) =
turn(s) and not(j = i) by concatenating them with conjunctions, substitut-
ing k with i because of the equation i = k, and deleting the tautology i = i.
p2 is obtained from the proposition,

Some contradiction may be found in a set of equations that characterize a
case even when CafeOBJ does not return false in a proof passage corresponding
to the case. If that is the case, a state predicate can be obtained from the
contradiction such that the state predicate does not hold in the case and is
likely invariant wrt an OTS concerned.

Let us consider the proof passage corresponding to case 2.exit.1.0.0.1 shown
in Example 3. We notice that the three equations pc(s,k) = cs, pc(s,i) = cs
and (i = k) = false contradict ∀υ : RSTlock .MX(υ) and inv1(s,i,k) can be
used in the proof passage.

Even when any contradictions are not found in a set of equations that charac-
terize a case and CafeOBJ does not return false in a proof passage correspond-
ing to the case, a state predicate may be found such that the state predicate can
be used to discharge the case and is likely invariant wrt an OTS concerned.

Let us consider the proof passage corresponding to case 1.check.1.1.0.
CafeOBJ returns pc(s,j) = cs xor true for the proof passage, but inv2(s,j,
i) also reduces to pc(s,j) = cs xor true in the proof passage. Therefore,
inv2(s,j,i) can be used to discharge the case and it is not necessary to split
the case anymore.

4 Soundness of Proof Scores

Let us consider the proof of ∀υ : RS . (p1(υ) ∧ . . . ∧ pN (υ)) described in Sub-
sect. 2.4 again. If CafeOBJ returns true for each proof passage in the proof
scores, p1, . . . , pN are really invariant wrt S provided that

1. Needless to say, the computer (including the operating system, the hardware,
etc.) on which CafeOBJ works is reliable,

2. Equational reasoning is sound and rewriting faithfully (partially though) im-
plements equational reasoning [8]; the CafeOBJ implementation of rewriting
is reliable,



3. The Hsiang TRS is sound [12]; the TRS is reliably implemented in CafeOBJ,
4. The built-in equality operator _==_ is not used,
5. S is specified in CafeOBJ in the way described in Subsect. 2.3, and
6. The proof scores of ∀υ : RS . (p1(υ) ∧ . . . ∧ pN (υ)) are written in the way

described in Subsect. 2.4.

When CafeOBJ meets the term a == b, it first reduces a and b to a′ and b′, which
are irreducible wrt a set of equations (rewrite rules) concerned, and returns true
if a′ is exactly the same as b′ and false otherwise. The combination of _==_ and
not_ can damage the soundness. Since the built-in inequality operator _=/=_ is
the combination of _==_ and not_, it should not be used either. Let us consider
the following module:

mod! DATA { [Data]

ops d1 d2 : -> Data

}

We try to prove ∀d : Data.¬(d = d2) by writing a proof score. A plausible proof
score that consists of one proof passage is as follows:

open DATA

op d : -> Data . -- an arbitrary value of Data.

red not(d == d2) . -- or red d =/= d2 .

close

CafeOBJ returns true for this proof passage, which contradicts the fact that
there exists the counterexample d2. Therefore, users should declare an equality
operator such as _=_ for each visible sort and equations defining it instead of
_==_ and _=/=_.

Under the above six assumptions, the only thing that we should take care of
on the soundness is whether all necessary cases are checked by rewriting for each
proof passage. A possible source of damaging it is transitions. Since transitions
are functions on states in OTSs, however, the source can be dismissed. Every
operator is a function in CafeOBJ as well. Therefore, rewriting surely covers all
necessary cases for each proof passage.

Note that we do not have to assume that the CafeOBJ specification of S,
when it is regarded as a TRS, is terminating or confluent for the soundness.
If the CafeOBJ specification is not terminating, CafeOBJ may not return any
results for a proof passage forever. This causes the success in proofs, but does
not affect the soundness.

We suppose that a term a has two irreducible forms a′ and a′′ in a proof
passage because the CafeOBJ specification is not confluent and that a actually
reduces to a′ but not to a′′. Although CafeOBJ ignores a rewriting sequence
that starts with a and ends in a′′, this does not affect the soundness because
a′ equals a′′ from an equational reasoning point of view and it is enough to use
either a′ or a′′. Whether the CafeOBJ specification is confluent, however, can
affects the success in proofs. Let us consider the following module:



mod! DATA2 { [Data2]

ops d1 d2 d3 : -> Data2

op _=_ : Data2 Data2 -> Bool {comm}

var D : Data2

eq (D = D) = true .

eq d1 = d2 . eq d1 = d3 .

}

We try to prove d1 = d3 by writing a proof passage. The case is split into two
(sub-)cases where (1) d2 = d3 and (2) d2 6= d3. Then, the proof score that
consists of two proof passages is as follows:

open DATA2

eq d2 = d3 .

red d1 = d3 .

close

open DATA2

eq (d2 = d3) = false .

red d1 = d3 .

close

CafeOBJ returns true for the first proof passage and false for the second proof
passage. We stuck for the second proof passage unless we notice the equation d1
= d3 in the module DATA2.

From what has been described, it is desirable that the CafeOBJ specification
of S is terminating and confluent.

We can check if proof scores that state predicates are invariant wrt S con-
forms to what is described in Subsect. 2.4. We suppose that all proofs are con-
ducted by simultaneous induction. Let P and P ′ be sets of state predicate such
that P ′ is empty. A procedure that makes such a check is as follows:

1. If P is empty, the procedure successfully terminates, which means that the
proof score of ∀υ : RS . p(υ) for each p ∈ P ′ conforms to what is described
in Subsect. 2.4; otherwise, extract a predicate p from P and go next.

2. Check if a proof score of ∀υ : RS . p(υ)q has been written. If so, go next;
otherwise, the procedure reports that a proof score of ∀υ : RS . p(υ) has not
been written and terminates.

3. Check if the proof score of ∀υ : RS . p(υ)q conforms to simultaneous induc-
tion. If so, go next; otherwise, the procedure reports that the proof score of
∀υ : RS . p(υ)q does not conform to simultaneous induction and terminates.

4. Check if the proof score of ∀υ : RS . p(υ)q covers all necessary cases. If so,
put p into P ′, put other state predicates that are used in the proof score
and that are not in P ′ into P , and go to 1; otherwise, the procedure reports
that the proof score of ∀υ : RS . p(υ) does not cover all necessary cases and
terminates.

The procedure can increase the confidence in soundness of proof scores.



5 Conclusion

We have described some tips on writing proof scores in the OTS/CafeOBJ
method and used Tlock, a mutual exclusion protocol using atomicInc, to ex-
emplify the tips. We have also informally argued soundness of proof scores in
the OTS/CafeOBJ method.

We have been developing a tool called Gateau [21] that takes propositions
used for case splitting and state predicates used to strengthen the basic induction
hypothesis, and generates the proof score of an invariant, which conforms to what
is described in Subsect. 2.4.

Proof scores can also be considered proof objects, which can be checked as
described in Sect. 4. We think that it is worthwhile to develop a tool, which is an
implementation of the procedure in Sect. 4 that checks if a proof score conforms
to what is described in Subsect. 2.4. Such a tool can be complementary to Gateau.

